

## Looking at the Big Picture – Snapshots of Patient Health in SAS®

Ruth Kurtycz, Spectrum Health – Healthier Communities, Grand Rapids, MI

### ABSTRACT

Healthcare professionals are exceedingly familiar with the multitude of measurements, diagnostics, and categorizations, which are used in the attempt to effectively define overall health. Commonly seen are measures such as Body Mass Index (BMI), cholesterol level, smoking status, activity level, etc. Yet each measurement itself only communicates a sliver in the big picture question of “how healthy am I?” To help see the big picture, the %REPORT macro has been developed which allows the user to take a “Snapshot” of a particular patient’s health at any given time. These snapshots are one-page reports that contain the most recent health diagnostics of interest for a particular patient and the dates they were taken. By displaying these measures together, healthcare providers can gain a more complete understanding of a patient’s overall health. The reports also include four visual displays of common health measurements over time, which allow the patient and the healthcare provider to identify improvements in health and together create action plans to address negative trends. To utilize the methods discussed here, users should have a basic understanding of the REPORT and GREPLAY procedures and access to SAS v9.3 or greater.

### INTRODUCTION

Personal health has become a huge industry. More and more people are interested in exploring how healthy they really are. The challenge for healthcare providers lies in how to accurately depict an complete view of a patient’s overall health. Commonly reported health measures such as BMI and blood pressure offer one perspective, but more specialized results, such as A1C levels for diabetics, can widen that perspective and provide better understanding for the patient. Furthermore adding lifestyle demographics such as smoking status, activity level, and diet quality can offer information not only on the current health of a patient, but also on their future wellbeing.

The snapshot program simultaneously reports multiple relevant measurements of patient health alongside visual displays to showcase a patient’s health trends. Although there are a multitude of measurements that could be included in these reports, the measures included here are BMI, LDL cholesterol, systolic and diastolic blood pressure, A1C, and fasting blood sugar. BMI, also known as body mass index is a screening tool for estimating unhealthy body weights calculated by dividing a patient’s weight in kilograms by their height in meters squared. Ratios above 25 are considered overweight while those over 30 are considered obese (CDC, 2015). Low-density lipoproteins (LDL) makes up the majority of the body’s cholesterol. Having high levels of LDL (more than 130 mg/dl) can lead to a buildup in the arteries and result in heart disease (CDC, 2015). Blood pressure is the force of blood pushing against the walls of the arteries. Having elevated blood pressure can permanently damage the heart, leading to heart disease and heart attacks if left unchecked (CDC, 2015). Blood pressure is measured with two separate numbers, systolic and diastolic blood pressure. Systolic blood pressure is the measure of force as the heart beats; it is considered high if it’s over 140 mmHg (CDC, 2015). Diastolic blood pressure is the measure of force between heart beats and is considered high if it’s over 90 mmHg (CDC, 2015). A1C is used to monitor controlled diabetes; the test estimates a patient’s average blood sugar over the last several months. A1C levels are considered high if they are over 6.5% (Mayo Clinic, 2015). Fasting blood sugar measures blood sugar levels after a patient has had no food for at least eight hours. It is useful in screening for diabetes. Measurements over 120 mg/dl are considered high (Mayo Clinic, 2015). When taken together, these laboratory diagnostics can begin to give us an idea of how a patient’s body is functioning at any given time.

For the lifestyle statistics, the reports include smoking status, weekly activity level, and daily servings of fruit and vegetables. Anti-smoking campaigns from all over the world proclaim the dangers of smoking, including but not limited to lung cancer, skin cancer, and heart disease (The Dangers of Smoking Cigarettes, 2012). Given the serious and sometimes fatal consequences of smoking, patient smoking status is an extremely relevant indicator of future health. In regards to activity level, studies have shown that regular physical activity improves mental and physical health while helping to prevent a variety of chronic diseases. The Center of Disease Control and Prevention recommends adults aged 18 to 65 aim for moderate intensity aerobic activity on five days a week for at least 30 minutes each time (CDC, 2015). Along with exercise, maintaining a healthy diet rich in fresh fruits and vegetables can provide the body with many essential vitamins and minerals which help prevent cancer. For a sedentary adult, the CDC recommends at least five servings of fruit and vegetables a day to meet the body’s nutritional needs. Displaying a patient’s activity level and their daily fruit and vegetable servings can help healthcare providers encourage preventative health and positive lifestyle changes. By combining this long term picture of a patient’s daily lifestyle with the “snapshot” of their bodily health provided by the laboratory measures (such as BMI

and blood pressure) mentioned above, health care providers can begin to attain a wider perspective on the overall view of patient health.

### DATA OVERVIEW

This program was originally created for use with the Core Health program data from Spectrum Health Healthier Communities in Grand Rapids, Michigan. Core Health is a free, 12-month program for adults with diabetes or heart failure. Patients in the Core Health program are provided with one-on-one home visits with a registered nurse and community health worker to help them better understand, manage, and improve their condition (Spectrum Health, 2015).


In order to protect patient privacy, the snapshot reports displayed in this paper were created using fabricated patients. This data is available for viewing and use in appendix 1. Besides the laboratory and lifestyle measurements discussed in the previous section, the patient dataset also includes ID, patient last name, patient first name, date of enrollment, date of visit, primary diagnosis, birth date, age in years, height in inches, weight in pounds, date of latest LDL cholesterol test, date of latest A1C test, name of registered nurse, location, and date of program completion. ID is a unique numerical variable assigned to each patient. Primary diagnosis is a numeric variable which has a value 1, 2, or 3 depending on whether the diagnosis is heart failure (1), diabetes (2), or a combination of both (3).

ID	Lastname	firstname	Enrollment	Visit	Diagnosis	DOB	age	Height	Weight	BMI	LDL	Choldate	SBP	DBP	A1C	A1Cdate	smoked	activity	fruitveg	BS	RN	location	closedate
1	Black	Aaron	1/1/2014	1/1/2014	2	3/20/1960	55	64.7	158.9	26.7	154	12/1/2013	136	100	6.0%	12/10/2013	yes	2	2	146	Hilary	United	12/12/2014
1	Black	Aaron	1/1/2014	2/3/2014	2	3/20/1960	55	64.7	161.0	27.1	154	12/1/2013	134	105	6.0%	12/10/2013	yes	5	4	130	Hilary	United	12/12/2014
1	Black	Aaron	1/1/2014	3/1/2014	2	3/20/1960	55	64.7	161.3	27.1	154	12/1/2013	125	103	6.0%	12/10/2013	yes	2	1	156	Hilary	United	12/12/2014
1	Black	Aaron	1/1/2014	4/10/2014	2	3/20/1960	55	64.7	164.6	27.7	152	3/1/2013	176	141	6.0%	12/10/2013	yes	1	5	114	Hilary	United	12/12/2014
1	Black	Aaron	1/1/2014	5/8/2014	2	3/20/1960	55	64.7	168.2	28.3	152	3/1/2013	158	130	6.0%	12/10/2013	yes	1	3	55	Hilary	United	12/12/2014
1	Black	Aaron	1/1/2014	6/20/2014	2	3/20/1960	55	64.7	168.5	28.3	152	3/1/2013	176	131	5.9%	8/5/2014	yes	2	4	141	Hilary	United	12/12/2014
1	Black	Aaron	1/1/2014	7/15/2014	2	3/20/1960	55	64.7	171.2	28.8	152	3/1/2013	178	150	5.9%	8/5/2014	yes	0	4	66	Hilary	United	12/12/2014
1	Black	Aaron	1/1/2014	8/31/2014	2	3/20/1960	55	64.7	166.8	28.0	193	7/1/2014	154	117	5.9%	8/5/2014	yes	1	1	149	Hilary	United	12/12/2014
1	Black	Aaron	1/1/2014	10/1/2014	2	3/20/1960	55	64.7	170.9	28.7	193	7/1/2014	182	161	5.9%	8/5/2014	yes	1	1	120	Hilary	United	12/12/2014
1	Black	Aaron	1/1/2014	11/12/2014	2	3/20/1960	55	64.7	172.5	29.0	193	7/1/2014	112	70	5.9%	8/5/2014	yes	4	3	166	Hilary	United	12/12/2014

Display 1. Generated Sample Code for Patient 1

### THE SNAPSHOT %REPORT MACRO

The snapshot report contains numerical measures of patient health, demographic information, and graphical displays. An outline of a snapshot report is displayed below. Although this program could be easily modified to graph a variety of data, this program is set to create graphs for physical activity, fruit and vegetable servings, BMI, and blood pressure (both systolic and diastolic). The patient data that will display on the report includes name, date of birth, diagnosis, enrollment date, latest visit date, close date (if applicable), location, RN name, and smoking status. Laboratory information includes LDL and date of last LDL test, A1C and date of last A1C test (if applicable), and blood sugar.



Display 2. Outline of Sample Snapshot Report

To make utilizing this snapshot reporting technique as quick and efficient as possible, a macro has been built which will build these reports based on the specifications of the parameters. This paper will highlight parts of the program code that are particularly important or complicated; the full code for this macro is available for viewing and use in appendix 3. Prior to the macro code, the program defines a custom format that associates the numeric diagnosis variable with the names of the diseases. Below the format, the patient data is read in from the work library and a patient full name variable is created from a concatenation of the patient's first and last names.

```
PROC FORMAT;
  VALUE diagnosis
    1      =      "Heart Failure"
    2      =      "Diabetes      "
    3      =      "HF/Diabetes  ";
RUN;

DATA pages;
  SET work.sample_data;
  Patient = CAT(TRIM(first_name)," ",TRIM(last_name));
RUN;
```

The %REPORT macro will generate individual snapshot reports for all patients in the pages dataset that meet the parameter specifications. This macro has three required parameters, dataset, output type, and output location, and four optional restriction parameters based on the values of ID, location, nurse name, and number of visits. Dataset specifies the SAS dataset where the patient data is located. Either an 'RTF' or 'PDF' output type should be specified; this allows the user to choose whether the snapshot reports are outputted as RTF or PDF files. Output location should specify the path where the final reports will be stored. The final reports will be stored either as RTF or PDF files with the patient's name in the file name.

```
%MACRO REPORT(dataset, output_type, output_location, ID_res, location_res,
  RN_res, visit_res);
```

## CHOOSING THE PATIENTS

An important part of the functionality of this program was creating the macro's ability to restrict which patients the snapshot reports are created for. At Healthier Communities, the Core Health program database contains visit information for thousands of patients; since creating a unique one page report for each of these patients would waste a massive amount of time and resources it is very important that only patients of interest are included in the data when the snapshot reports are built. The four restriction parameters in the macro allow the patient dataset to be restricted by patient ID number, patient location, the name of the registered nurse overseeing the patient's checkup, and number of visits a patient has had.

The macro executes a series of %IF, %THEN, and %DO statements to restrict the patient dataset to the specified restriction parameters. When a valid patient ID is specified for the ID restriction parameter all observations excluding those with a matching ID number are removed from the data. Using the newly created restrict dataset, similar code is executed in turn for the location and RN restriction parameters. All the restriction parameters are optional so if a value is not defined the dataset is left intact.

```
%MACRO REPORT(dataset, output_type, output_location, ID_res, location_res,
  RN_res, visit_res);
...
%IF &ID_res^=. %THEN %DO;
  DATA restrict;
    SET &dataset;
    WHERE ID=&ID_res;

  RUN;
%END;
%ELSE %DO;
  DATA restrict;
    SET &dataset;

  RUN;
%END;
...
%MEND;
```

The final parameter is used to restrict the creation of snapshot reports only to patients who have at least a certain number of visits. This is an important consideration with regards to the graphical displays since patients with very few visits create graphs which are visually unappealing. The visit restriction can be any positive number but it will default to six when unspecified or negative. The restriction dataset is sorted by last name and ID; this was done in the original dataset because patients may have multiple enrollments in the Core Health program, giving some patient ID's more than one enrollment date. The count variable tracks how many visits a patient has for each unique enrollment date. When the final count is greater than or equal to the visit restriction parameter the use variable is assigned a value of one. The sort procedure below creates a dataset which contains a list of the acceptable patient ID's and last names.

```
%MACRO REPORT(dataset, output_type, output_location, ID_res, location_res,
RN_res, visit_res);
...
%IF &visit_res=. %THEN %LET visit_res=6;
%IF &visit_res<0 %THEN %LET visit_res=6;
DATA restrict;
    SET restrict;
    BY last_name ID;
    count+1;
    IF FIRST.ID THEN count=1;
    IF count>=&visit_res THEN use=1;
RUN;
PROC SORT DATA=restrict(WHERE=(use=1)) NODUPKEY
OUT=use(KEEP=ID use last_name);
    BY last_name ID;
RUN;
...
%MEND;
```

The use dataset is merged with the restrict dataset to ensure that all visits within an acceptable patient ID are flagged to be used. The resulting final dataset contains only the patients who meet the restriction parameter criteria. All patients who do not have at least the minimum acceptable number of visits are deleted from the final dataset. The variable N tracks the total number of patients for whom a snapshot report will be created.

```
%MACRO REPORT(dataset, output_type, output_location, ID_res, location_res,
RN_res, visit_res);
...
DATA final;
    RETAIN N;
    MERGE restrict(DROP=use) use;
    BY last_name ID;
    IF use^=1 THEN DELETE;
    IF FIRST.ID THEN N+1;
    DROP use;
RUN;
...
%MEND;
```

Using PROC MEANS the macro finds the maximum number of reports that will be created. It stores this value in the macro variable maxp using the CALL SYMPUT statement; this value dictates how many times the report creation process will reiterate.

```
%MACRO REPORT(dataset, output_type, output_location, ID_res, location_res,
RN_res, visit_res);
...
PROC MEANS DATA=final MAX NOPRINT;
    VAR N;
    OUTPUT OUT=maxs(DROP= _TYPE_ _FREQ_) MAX=maxp;
RUN;
DATA _NULL_;
    SET maxs;
    CALL SYMPUT ('maxp', INPUT (maxp, BEST. ) );
```

```
RUN;
...
%MEND;
```

## PATIENT DATASETS AND MACRO VARIABLES

To ensure that the correct values are being printed on each snapshot report, a series of macro variables were used for each of the health measurements. These macro variables are unique to each patient and assigned prior to the creation of the reports. The %DATAS macro below breaks down the final dataset to create a unique dataset for each patient which contains only the observations related to an individual patient. The macro then uses the MEANS procedure to locate the largest visit date within the data for each individual patient. This date is outputted to the macro variable &maxvisits using a method similar to the assignment of the total patient macro variable seen in the previous section. This macro variable is used to create a latest visit date variable within each patient dataset.

```
%MACRO REPORT(dataset, output_type, output_location, ID_res, location_res,
RN_res, visit_res);
...
%MACRO DATAS(maxp);
%DO i=1 %TO &maxp;
    DATA patient&i;
        SET final;
        WHERE N=&i;
    RUN;
    PROC MEANS DATA=patient&i MAX NOPRINT;
        VAR visit;
        OUTPUT OUT=maxvisits&i(DROP= _TYPE_ _FREQ_) MAX=maxvd;
    RUN;
    DATA _NULL_;
        SET maxvisits&i;
        CALL SYMPUT ('maxvd', INPUT (maxvd, BEST.) );
    RUN;
    DATA patient&i;
        SET patient&i;
        IF _N_>=1 THEN maxvd=&maxvd;
        FORMAT maxvd MMDDYY10.;
    RUN;
...
%END;
%MEND; %DATAS (&maxp);
...
%MEND;
```

The %OUTPUT macro assigns the values of the relevant health and demographic variables to other macro variables using the CALL SYMPUT method used in assignment of the latest visit macro variable above. Macro variables are created for patient name, birthdate, diagnosis (both the numerical and the character values), enrollment date, date of completion, most recent visit date, RN name, A1C result, A1C test date, LDL result, LDL test date, smoking status, location, blood sugar, and a macro variable containing today's date.

```
%MACRO REPORT(dataset, output_type, output_location, ID_res, location_res,
RN_res, visit_res);
...
%MACRO OUTPUT(maxp);
%DO i=1 %TO &maxp;
    DATA _NULL_;
        SET patient&i;
        today=date();
        CALL SYMPUT ('patient', PUT (patient, $30.) );
        ...
        CALL SYMPUT ('today', PUT (today, mmddyy10.));
    RUN;
...
%MEND;
...
%MEND;
```

## PATIENT HEALTH TABLE

The top of the snapshot report shows demographics on each patient and numerical measures of their health. This serves as an ideal location to display health measures who are not suited to graphical displays, but nevertheless still important in creating an overview of patient health. To achieve the correct alignment in the text, the REPORT procedure was used to create the visible output. Four text datasets were created, each of which contained one line of the text to be displayed in the final patient health table at the top of each report.

In the code below, the first of these four datasets is displayed. The first line of the snapshot report will contain the macro variable values for patient name, date of birth, RN name, and location, along with the static text to label each of these values. ODS ESCAPECHAR allows use of the available inline formatting functions to change the style within cells, text, titles, and footnotes (SAS.com, 2015). With the '#' defined as the escape character, the macro was able to dictate the appearance of the text, including the size and the font weight. Once the text variables were defined, the RESOLVE, QUOTE and DEQUOTE statements were used to ensure the macro variables and inline formatting were properly displayed in the report output.

```
%MACRO REPORT(dataset, output_type, output_location, ID_res, location_res,
RN_res, visit_res);
...
ODS ESCAPECHAR = "#";
...
%MACRO OUTPUT(maxp);
...
DATA titles1&i;
  w="#S={OUTPUTWIDTH=25% FONTWEIGHT=BOLD}Patient Name: #S={}&patient";
  x="#S={OUTPUTWIDTH=40% FONTWEIGHT=BOLD}Date of Birth: #S={}&mDOB";
  y="#S={OUTPUTWIDTH=10% FONTWEIGHT=BOLD}RN: #S={}&RN";
  z="#S={OUTPUTWIDTH=15% FONTWEIGHT=BOLD just=1}Location: #S={}&location";
  wResolved=DEQUOTE (RESOLVE (QUOTE (w) ) ) ;
  xResolved=DEQUOTE (RESOLVE (QUOTE (x) ) ) ;
  yResolved=DEQUOTE (RESOLVE (QUOTE (y) ) ) ;
  zResolved=DEQUOTE (RESOLVE (QUOTE (z) ) ) ;
RUN;
...
%MEND;
...
%MEND;
```

## CONDITIONAL OUTPUT

There were several variables in the Core Health dataset that were only of interest to specific segments of the patient population, such as the A1C value which is only applicable to diagnosed diabetics. To avoid printing irrelevant health statistics for the heart failure patients, some of the patient health table datasets conditionally executed based on the diagnosis of the patient. When the patient diagnosis was either a 2 or 3, corresponding to a diagnosis of either diabetes or a combination of heart failure and diabetes, the A1C test result and the date of the latest A1C test are displayed on the third line of the report. When the diagnosis is 1, corresponding to a diagnosis of heart failure, smoking status is displayed on the third line of the report instead. Smoking status is also displayed in the snapshot reports for diabetic patients but it is on the fourth line of the table, below the A1C results. After the four patient health table datasets are created, they are merged to create one dataset which contains all of the text to be included in the REPORT procedure output.

```
%MACRO REPORT(dataset, output_type, output_location, ID_res, location_res,
RN_res, visit_res);
...
%MACRO OUTPUT(maxp);
...
%IF &Diagnosis>1 %THEN %DO;
  DATA titles2&i;
    ...
  RUN;
  DATA titles3&i;
    ...
%END;
```

```

y="#"#S={OUTPUTWIDTH=10% FONTWEIGHT=BOLD}A1C: #S={}&A1C";
z="#"#S={OUTPUTWIDTH=15% FONTWEIGHT=BOLD just=1}
A1C Date: #S={}&A1Cd";
...
yResolved=DEQUOTE (RESOLVE (QUOTE (y) ));
zResolved=DEQUOTE (RESOLVE (QUOTE (z) ));
RUN;
DATA titles4&i;
...
RUN;
%END;
%ELSE %DO;
...
%END;
DATA titles&i;
    SET titles1&i titles2&i titles3&i titles4&i;
RUN;
...
%MEND;
...
%MEND;

```

### PROC REPORT

The output from the REPORT generally displays as a formatted table. Since the snapshot reports only use PROC REPORT to align the text, the STYLE and NOHEADER options were invoked to repress the table border and title. Lastly, the four text variables are set to DISPLAY so they appear correctly in the final table. In the snapshot program the PROC REPORT code above appears in both the RTF and PDF output but there is no difference in the syntax.

```

%MACRO REPORT(dataset, output_type, output_location, ID_res, location_res,
RN_res, visit_res);
...
%MACRO OUTPUT(maxp);
...
PROC REPORT DATA=titles&i NOWD STYLE(REPORT)=[BORDERCOLOR=white BORDERWIDTH=0]
NOCENTER NOHEADER;
    COLUMN wResolved xResolved yResolved zResolved;
    DEFINE wResolved/DISPLAY;
    DEFINE xResolved/DISPLAY;
    DEFINE yResolved/DISPLAY;
    DEFINE zResolved/DISPLAY;
RUN;
...
%MEND;
...
%MEND;

```

### PATIENT HEALTH GRAPHICS

The visual displays of patient health are the most important tool health care providers have when working with patients on improving their health. Graphics, unlike numbers and tables, are less intimidating for the average patient to view and much more intuitive in their understanding. The macro outputs four graphs of patient health measure over time. As stated earlier, the measures chosen for display in the graphs are patient BMI, fruit and vegetable servings, activity level, and blood pressure. The macro gives the option to produce either RTF or PDF reports.

### CHOOSING RTF OR PDF OUTPUT

When the user specifies an output type the code below creates either the PDF or RTF report and stores it in the output location with each report saved in a file named "Snapshot\_(patient name)." Although patient name is used for ease of identification, ID number could be easily substituted for increased confidentiality. Use of the STARTPAGE=NEVER overrides the command to start a new page for each new graphic, which is essential for keeping the snapshots report to one page

```
ODS &output_type FILE="&output_location/SNAPSHOT_&patient.&output_type"
```

```
STARTPAGE=NEVER;
```

## THE GPLOT AND GSLIDE PROCEDURES

The output uses a graphics library to store the graphs before the GREPLAY template specifies their arrangement in the report. To eliminate potential errors in the recall of graphs from this library, the %DELCAT macro below is used to clear the library before the new graphs are created. This macro will run when the RTF output is specified, regardless of the previous contents of the graphics library.

```
%MACRO REPORT(dataset, output_type, output_location, ID_res, location_res,
RN_res, visit_res);
...
%MACRO OUTPUT(maxp);
...
%IF &output_type=RTF %THEN %DO;
    %MACRO DELCAT(catname);
        PROC GREPLAY NOFS IGOUT=&catname;
            DELETE _ALL_;
        RUN;
        QUIT;
    %MEND;
    %DELCAT(work.graphics);
...
%END;
...
%MEND;
...
%MEND;
```

The snapshot reports use graphs originally created by the GPLOT procedure. For each GPLOT graph, the visit date and the variable of interest are labeled to make the final graph easier to understand. The GOUT statement stores the created graphs in the graphics library so they will be accessible for the GREPLAY template to access later. The NAME option in the PLOT statement identifies each graph within the library. The BMI, fruit/veggie, and activity graphics are similar in their creation but the blood pressure graph uses the OVERLAY option to display both diastolic and systolic blood pressure on the same graph.

```
%MACRO REPORT(dataset, output_type, output_location, ID_res, location_res,
RN_res, visit_res);
...
%MACRO OUTPUT(maxp);
...
    GOPTIONS RESET=ALL NODISPLAY HZIZE=4in VSIZE=1.75in;
    SYMBOL1 INTERPOL=JOIN VALUE=PLUS HEIGHT=3;
    PROC GPLOT DATA=patient&i GOUT=graphics;
        LABEL Visit="Visit date";
        LABEL BMI="BMI";
        PLOT BMI*Visit/NAME="BMI&i" NOFRAME;
    RUN;
    '''
    PROC GPLOT DATA=patient&i GOUT=graphics;
        LABEL Visit="Visit date";
        LABEL SBP="BP";
        PLOT SBP*Visit DBP*Visit/NAME="BP&i" OVERLAY NOFRAME;
    RUN;
    ...
    %END;
...
%MEND;
...
%MEND;
```

Next the GSLIDE procedure is used to create titles for each of the graphs. This is done because the titles automatically created for the GPLOT graphs are lost when they are stored in the graphics library. Like the graphs, each slide is given a name and stored in the graphics library using the GOUT option. Each slide contains only a title



centered across the top of the slide; the graphs will be positioned over the slides in the GREPLAY template so the titles display directly above each graph.

```
%MACRO REPORT(dataset, output_type, output_location, ID_res, location_res,
RN_res, visit_res);
...
%MACRO OUTPUT(maxp);
...
    PROC GSLIDE GOUT=graphics NAME="BMItitle&i";
    NOTE HEIGHT=3.5 BOLD JUSTIFY=CENTER "Body Mass Index";
    RUN;
    PROC GSLIDE GOUT=graphics NAME="Activetitle&i";
    NOTE HEIGHT=3.5 BOLD JUSTIFY=CENTER
    "Average Times Physically Active Per Week";
    RUN;
    PROC GSLIDE GOUT=graphics NAME="Fruitttitle&i";
    NOTE HEIGHT=3.5 BOLD JUSTIFY=CENTER
    "Average Fruit/Veggie Servings Per Day";
    RUN;
    PROC GSLIDE GOUT=graphics NAME="Bptitle&i";
    NOTE HEIGHT=3.5 BOLD JUSTIFY=CENTER
    "Systolic and Diastolic Blood Pressure";
    RUN;
...
%END
...
%MEND;
...
%MEND;
```

## PROC GREPLAY

PROC GREPLAY defines a custom template, which has space for all four graphics, and their title slides. The TEDF statement names the custom template "snapshot" and stores it in the tempcat work library as specified in the TC option. Each of the eight display areas in the template are sized and positioned by specifying x and y axis values for the four corners (lower left, upper left, lower right, and upper right).

```
%MACRO REPORT(dataset, output_type, output_location, ID_res, location_res,
RN_res, visit_res);
...
%MACRO OUTPUT(maxp);
...
    GOPTIONS RESET=ALL DISPLAY HSIZE=9in VSIZE=4.3in;
    PROC GREPLAY TC=work.tempcat NOFS;
        TEDF snapshot
        1/LLX=0 LLY=0 ULX=0 ULY=44
          urx=49 ury=44 lrx=49 lry=0
        2/llx=0 lly=0 ulx=0 uly=49
          urx=49 ury=49 lrx=49 lry=0
        3/llx=0 lly=51 ulx=0 uly=95
          urx=49 ury=95 lrx=49 lry=51
        4/llx=0 lly=51 ulx=0 uly=100
          urx=49 ury=100 lrx=49 lry=51
        5/llx=51 lly=51 ulx=51 uly=95
          urx=100 ury=95 lrx=100 lry=51
        6/llx=51 lly=51 ulx=51 uly=100
          urx=100 ury=100 lrx=100 lry=51
        7/llx=51 lly=0 ulx=51 uly=44
          urx=100 ury=44 lrx=100 lry=0
        8/llx=51 lly=0 ulx=51 uly=49
          urx=100 ury=49 lrx=100 lry=0;
        TEMPLATE snapshot;
    QUIT;
%END;
```

```
...
%MEND;
...
%MEND;
```


Once the template is defined the GREPLAY procedure is again used to assign each of the slides and graphs stored in the graphics library to a location on the template. The IGOUT option specifies the library where the graphics are stored, the TMEPLATE option specifies the template, and the TC option identifies the location of the template. Lastly, the date modified is added to the report underneath the graphics output using ODS TEXT. The snapshot report is now complete.

```
%MACRO REPORT(dataset, output_type, output_location, ID_res, location_res,
RN_res, visit_res);
...
%MACRO OUTPUT(maxp);
...
PROC GREPLAY IGOUT=graphics NOFS TC=work.tempcat TEMPLATE=newtemp;
TREPLAY 1:"Fruit&i" 2:"Fruittitle&i" 3:"BMI&i" 4:"BMItitle&i"
5:"Active&i" 6:"Activetitle&i" 7:"BP&i" 8:"BPTitle&i";
RUN;
QUIT;
ODS TEXT = "#S={OUTPUTWIDTH=100% JUST=R FONTWEIGHT=BOLD}
Date Modified: #S={}&today ";
ODS &output_type CLOSE;
%END;
...
%MEND;
...
%MEND;
```

## SNAPSHOT REPORT

Below is the final output of a sample snapshot report. Along the top the PROC REPORT output containing patient information, dates, and relevant lab information is displayed. Below are the four graphs and the date modified. Glancing over the report, one can easily assess a patient's overall health, identify trends, and quickly locate areas that need improvement. It would be very easy for a health care provider to sit down with a patient and go through each area of the report, giving them verbal and visual explanations of their status without overwhelming the patient with information.

Additionally, this reporting method has applications outside of those discussed in this paper. The reports could easily be created for groups of patients with similar traits of interest rather than for individuals, such as all patients with a particular diagnosis or from the same treatment center. In fact there is no reason snapshot reporting needs to be restricted to just the health care field; it could, for example, be used in retail industries to track the sales, margin, and quantity of a specific product, or statistics on a team or player's performance within the sports field. The opportunities for customization and application are virtually endless.



**Display 3. Final Snapshot Report Output for RTF (left) and PDF (right)**

### CONCLUSION

Snapshot reporting gives healthcare providers another useful technique for evaluating and explaining a patient's overall health. Within the endless measures and statistics for evaluating health, these reports provide an effective and informative display of relevant information in a way both patients and health care providers can intuitively grasp. By visualizing multiple health measures at once, both positive and negative trends can be easily identified and addressed before they become a problem. In short, snapshot reports can be a powerful tool for promoting increased patient understanding and teamwork between a patient and healthcare provider.

## REFERENCES

- CDC: Center for Disease Control and Prevention. (2015, February 23). Retrieved from <http://www.cdc.gov/>
- Mayo Clinic. (2015). A1C Test. Retrieved March 21, 2015, from <http://www.mayoclinic.org/tests-procedures/a1c-test/>
- SAS support. (2014). %MACRO statement. In SAS. Retrieved April, 2014, from <http://support.sas.com>
- Slaughter, S. J., & Delwiche, L. D. (2004). SAS macro programming for beginners. Retrieved April, 2014, from <http://www2.sas.com/proceedings/sugi29/243-29.pdf>
- Spectrum Health. (2015). Core Health Program. In Healthier Communities. Retrieved from <http://www.spectrumhealth.org/corehealth>
- Stroupe, J. (2003). Nine Steps to Get Started using SAS® Macros. Retrieved April, 2014, from <http://www2.sas.com/proceedings/sugi28/056-28.pdf>
- The Dangers of Smoking Cigarettes (2012). In Quit Smoking Central. Retrieved from <http://www.quit-smoking-central.com/dangers-of-smoking.html>







## APPENDIX 2: SNAPSHOT REPORTS FOR ALL SAMPLE PATIENTS

### Patient Report

**Patient Name:** Adam Orange  
**Diagnosis:** Diabetes  
**Enrollment Date:** 11/02/2014  
**Date:** .

**Date of Birth:** 12/31/1959  
**Latest Visit Date:** 02/01/2015

**RN:** Jamie      **Location:** Core Health  
**LDL:** .      **LDL Date:** .  
**A1C:** 5.2%      **A1C Date:** 10/01/2014  
**Smoker?:** yes      **Blood Sugar:** 125







Date Modified: 04/02/2015

### Patient Report

**Patient Name:** Amy Pink  
**Diagnosis:** HF/Diabetes  
**Enrollment Date:** 07/15/2014  
**Date:** .

**Date of Birth:** 10/12/1940  
**Latest Visit Date:** 01/05/2015

**RN:** Hilary      **Location:** United  
**LDL:** 207      **LDL Date:** 06/01/2014  
**A1C:** 7.0%      **A1C Date:** 06/11/2014  
**Smoker?:** no      **Blood Sugar:** 120







Date Modified: 04/02/2015

Patient Report

Patient Name: Melissa Red  
 Diagnosis: HF/Diabetes  
 Enrollment Date: 06/20/2014  
 Date:

Date of Birth: 05/16/1959  
 Latest Visit Date: 01/19/2015

RN: Hilary Location: United  
 LDL: . LDL Date: .  
 A1C: . A1C Date: .  
 Smoker?: yes Blood Sugar: 126



Date Modified: 04/02/2015





Patient Report



Patient Name: Anna Green  
 Diagnosis: Diabetes  
 Enrollment Date: 01/31/2014  
 Date: 01/13/2015

Date of Birth: 01/15/1954  
 Latest Visit Date: 11/01/2014

RN: Hilary Location: Core Health  
 LDL: 176 LDL Date: 01/02/2014  
 A1C: . A1C Date: .  
 Smoker?: no Blood Sugar: 156







Date Modified: 04/02/2015

*Patient Report*

**Patient Name:** Chris Blue  
**Diagnosis:** Heart Failure  
**Enrollment Date:** 04/05/2014  
**Close Date:** 01/20/2014

**Date of Birth:** 02/25/1972  
**Latest Visit Date:** 11/30/2014

**RN:** Jamie  
**LDL:**176  
**Smoker?:** no  
**Location:** Core Health  
**LDL Date:** 03/30/2014  
**Blood Sugar:** 160







Date Modified: 04/02/2015

*Patient Report*

**Patient Name:** Anna Green  
**Diagnosis:** Diabetes  
**Enrollment Date:** 01/31/2014  
**Date:** 01/13/2015

**Date of Birth:** 01/15/1954  
**Latest Visit Date:** 11/01/2014

**RN:** Hilary  
**LDL:** 176  
**A1C:** .  
**Smoker?:** no  
**Location:** Core Health  
**LDL Date:** 01/02/2014  
**Blood Sugar:** 156



Date Modified: 04/02/2015






*Patient Report*

**Patient Name:** Ruth Brown  
**Diagnosis:** Heart Failure  
**Enrollment Date:** 03/01/2014  
**Close Date:** 02/01/2015

**Date of Birth:** 07/30/1942  
**Latest Visit Date:** 12/03/2014

**RN:** Jamie      **Location:** Core Health  
**LDL:**            **LDL Date:** .  
**Smoker?:** yes  
**Blood Sugar:** 109



**Date Modified:** 04/02/2015

### APPENDIX 3: FULL PROGRAM CODE

```

PROC FORMAT;
VALUE diagnosis
    1      =      "Heart Failure"
    2      =      "Diabetes      "
    3      =      "HF/Diabetes  ";
RUN;

%let workdir=%trim(%sysfunc(pathname(work)));
DATA pages;
    SET work.sample_data;
    Patient = CAT(TRIM(first_name)," ",TRIM(last_name));
RUN;

%MACRO REPORT(dataset, output_type, output_location, ID_res, location_res, RN_res,
visit_res);

OPTIONS SYMBOLGEN ORIENTATION=LANDSCAPE NONUMBER NODATE NOBYLINE CENTER
TOPMARGIN=0.5in BOTTOMMARGIN=0.5in
LEFTMARGIN=0.5in RIGHTMARGIN=0.5in; TITLE;
ODS ESCAPECHAR = "#";

    %IF &ID_res^=. %THEN %DO;
        DATA restrict;
            SET &dataset;
            WHERE ID=&ID_res;
        RUN;
    %END;
    %ELSE %DO;
        DATA restrict;
            SET &dataset;
        RUN;
    %END;

    %IF &location_res= %THEN %DO;
        DATA restrict;
            SET restrict;
        RUN;
    %END;
    %ELSE %DO;
        DATA restrict;
            SET restrict;
            WHERE location="&location_res";
        RUN;
    %END;

    %IF &RN_res= %THEN %DO;
        DATA restrict;
            SET restrict;
        RUN;
    %END;
    %ELSE %DO;
        DATA restrict;
            SET restrict;
            WHERE RN="&RN_res";
        RUN;
    %END;

PROC SORT DATA=restrict;
    BY last_name ID;
RUN;

%IF &visit_res=. %THEN %LET visit_res=6;

```

## Looking at the Big Picture – Snapshots of Patient Health in SAS®

```

%IF &visit_res<0 %THEN %LET visit_res=6;
DATA restrict;
    SET restrict;
    BY last_name ID;
        count+1;
        IF FIRST.ID THEN count=1;
        IF count>=&visit_res THEN use=1;
RUN;

PROC SORT DATA=restrict(WHERE=(use=1)) NODUPKEY
OUT=use(KEEP=ID use last_name);
    BY last_name ID;
RUN;

DATA final;
    RETAIN N;
    MERGE restrict(DROP=use) use;
    BY last_name ID;
        IF use^=1 THEN DELETE;
        IF FIRST.ID THEN N+1;
    DROP use;
RUN;

PROC MEANS DATA=final MAX NOPRINT;
    VAR N;
    OUTPUT OUT=maxs(DROP= _TYPE_ _FREQ_) MAX=maxp;
RUN;
DATA _NULL_;
    SET maxs;
    CALL SYMPUT ('maxp', INPUT (maxp, BEST. ) );
RUN;

%MACRO DATAS(maxp);
%DO i=1 %TO &maxp;
    DATA patient&i;
        SET final;
        WHERE N=&i;
    RUN;
    PROC MEANS DATA=patient&i MAX NOPRINT;
        VAR visit;
        OUTPUT OUT=maxvisits&i(DROP= _TYPE_ _FREQ_) MAX=maxvd;
    RUN;
    DATA _NULL_;
        SET maxvisits&i;
        CALL SYMPUT ('maxvd', INPUT (maxvd, BEST.) );
    RUN;
    DATA patient&i;
        SET patient&i;
        IF _N_>=1 THEN maxvd=&maxvd;
        FORMAT maxvd MMDDYY10.;
    RUN;
    PROC SORT NODUPKEY DATA=patient&i;
        BY visit;
    RUN;
%END;
%MEND; %DATAS(&maxp);

%MACRO OUTPUT(maxp);
%DO i=1 %TO &maxp;
    DATA _NULL_;
    SET patient&i;
        today=date();
        CALL SYMPUT ('patient', PUT (patient, $30.) );

```

## Looking at the Big Picture – Snapshots of Patient Health in SAS®

```

CALL SYMPUT ('mDOB', PUT(DOB,mmdyy10.));
CALL SYMPUT ('mDiagnosis', PUT(Diagnosis, diagnosis.) );
CALL SYMPUT ('Diagnosis', INPUT(Diagnosis, Best.) );
CALL SYMPUT ('EnrollDate', PUT(Enrollment,mmdyy10.));
CALL SYMPUT ('CloseDate', PUT(closedate,mmdyy10.));
CALL SYMPUT ('mmaxvd', PUT(maxvd,mmdyy10.));
CALL SYMPUT ('RN', PUT(RN, $10.));
CALL SYMPUT ('A1C', PUT(A1C, Percent.1));
CALL SYMPUT ('A1Cd', PUT(A1C_date,mmdyy10.));
CALL SYMPUT ('LDL', PUT(compress(LDL), $3.));
CALL SYMPUT ('LDLd', PUT(CHOL_date,mmdyy10.));
CALL SYMPUT ('Smoked', PUT(Smoked, $5.));
CALL SYMPUT ('BS', PUT(compress(BS), $3.));
CALL SYMPUT ('location', PUT (location, $20.) );
CALL SYMPUT ('today', PUT(today,mmdyy10.));

RUN;

DATA titles1&i;
w="#S={OUTPUTWIDTH=25% FONTWEIGHT=BOLD}Patient Name: #S={}&patient";
x="#S={OUTPUTWIDTH=40% FONTWEIGHT=BOLD}Date of Birth: #S={}&mDOB";
y="#S={OUTPUTWIDTH=10% FONTWEIGHT=BOLD}RN: #S={}&RN";
z="#S={OUTPUTWIDTH=15% FONTWEIGHT=BOLD just=1}Location: #S={}&location";
wResolved=DEQUOTE (RESOLVE (QUOTE (w) ));
xResolved=DEQUOTE (RESOLVE (QUOTE (x) ));
yResolved=DEQUOTE (RESOLVE (QUOTE (y) ));
zResolved=DEQUOTE (RESOLVE (QUOTE (z) ));

RUN;

%IF &Diagnosis>1 %THEN %DO;
DATA titles2&i;
w="#S={OUTPUTWIDTH=25% FONTWEIGHT=BOLD}
Diagnosis: #S={}&mDiagnosis";
x="#S={OUTPUTWIDTH=40% FONTWEIGHT=BOLD} ";
y="#S={OUTPUTWIDTH=10% FONTWEIGHT=BOLD}LDL: #S={}&LDL";
z="#S={ OUTPUTWIDTH=15% FONTWEIGHT=BOLD JUST=L
}LDL Date: #S={}&LDLd";
wResolved=DEQUOTE (RESOLVE (QUOTE (w) ));
xResolved=DEQUOTE (RESOLVE (QUOTE (x) ));
yResolved=DEQUOTE (RESOLVE (QUOTE (y) ));
zResolved=DEQUOTE (RESOLVE (QUOTE (z) ));

RUN;

DATA titles3&i;
w="#S={OUTPUTWIDTH=25% FONTWEIGHT=BOLD}
Enrollment Date: #S={}&EnrollDate";
x="#S={OUTPUTWIDTH=40% FONTWEIGHT=BOLD}
Latest Visit Date: #S={}&mmaxvd";
y="#S={OUTPUTWIDTH=10% FONTWEIGHT=BOLD}A1C: #S={}&A1C";
z="#S={OUTPUTWIDTH=15% FONTWEIGHT=BOLD JUST=L}
A1C Date: #S={}&A1Cd";
wResolved=dequote (resolve (quote (w) ));
xResolved=dequote (resolve (quote (x) ));
yResolved=dequote (resolve (quote (y) ));
zResolved=dequote (resolve (quote (z) ));

RUN;

DATA titles4&i;
w="#S={OUTPUTWIDTH=25% FONTWEIGHT=BOLD}Date: #S={}&closedate";
x="#S={OUTPUTWIDTH=40% FONTWEIGHT=BOLD} ";
y="#S={OUTPUTWIDTH=10% FONTWEIGHT=BOLD}Smoker?: #S={}&Smoked";
z="#S={OUTPUTWIDTH=15% FONTWEIGHT=BOLD JUST=L}
Blood Sugar: #S={}&BS";
wResolved=dequote (resolve (quote (w) ));
xResolved=dequote (resolve (quote (x) ));
yResolved=dequote (resolve (quote (y) ));
zResolved=dequote (resolve (quote (z) ));

```

## Looking at the Big Picture – Snapshots of Patient Health in SAS®

```

RUN;
%END;
%ELSE %DO;
DATA titles2&i;
    w="#S={OUTPUTWIDTH=25% FONTWEIGHT=BOLD}
    Diagnosis: #S={}&mDiagnosis";
    x="#S={OUTPUTWIDTH=40% FONTWEIGHT=BOLD} ";
    y="#S={OUTPUTWIDTH=10% FONTWEIGHT=BOLD}LDL: #S={}&LDL";
    z="#S={OUTPUTWIDTH=15% FONTWEIGHT=BOLD JUST=L}
    LDL Date: #S={}&LDLd";
    wResolved=dequote(resolve(quote(w)));
    xResolved=dequote(resolve(quote(x)));
    yResolved=dequote(resolve(quote(y)));
    zResolved=dequote(resolve(quote(z)));
RUN;
DATA titles3&i;
    w="#S={OUTPUTWIDTH=25% FONTWEIGHT=BOLD}
    Enrollment Date: #S={}&EnrollDate";
    x="#S={OUTPUTWIDTH=40% FONTWEIGHT=BOLD}
    Latest Visit Date: #S={}&mmaxvd";
    y="#S={OUTPUTWIDTH=10% FONTWEIGHT=BOLD}Smoker?: #S={}&Smoked";
    z=" ";
    wResolved=dequote(resolve(quote(w)));
    xResolved=dequote(resolve(quote(x)));
    yResolved=dequote(resolve(quote(y)));
    zResolved=dequote(resolve(quote(z)));
RUN;
DATA titles4&i;
    w="#S={OUTPUTWIDTH=25% FONTWEIGHT=BOLD}
    Close Date: #S={}&closedate";
    x=" ";
    y=" ";
    z="#S={OUTPUTWIDTH=15% FONTWEIGHT=BOLD JUST=L}
    Blood Sugar: #S={}&BS";
    wResolved=dequote(resolve(quote(w)));
    xResolved=dequote(resolve(quote(x)));
    yResolved=dequote(resolve(quote(y)));
    zResolved=dequote(resolve(quote(z)));
RUN;
%END;

DATA titles&i;
    SET titles1&i titles2&i titles3&i titles4&i;
RUN;

ODS &output_type FILE="&output_location/SNAPSHOT_&patient.&output_type"
STARTPAGE=NEVER;

TITLE "#S={OUTPUTWIDTH=100% JUST=C FONTWEIGHT=BOLD}Patient Report";

PROC REPORT DATA=titles&i NOWD STYLE(REPORT)=[BORDERCOLOR=white
BORDERWIDTH=0] NOCENTER NOHEADER;
    COLUMN wResolved xResolved yResolved zResolved;
    DEFINE wResolved/DISPLAY;
    DEFINE xResolved/DISPLAY;
    DEFINE yResolved/DISPLAY;
    DEFINE zResolved/DISPLAY;
RUN;

%MACRO DELCAT(catname);
    PROC GREPLAY NOFS IGOUT=&catname;
    DELETE _ALL_;
RUN;

```

## Looking at the Big Picture – Snapshots of Patient Health in SAS®

```
QUIT;
%MEND;
%DELCAT(work.graphics);

GOPTIONS RESET=ALL NODISPLAY ;
SYMBOL1 INTERPOL=JOIN VALUE=PLUS HEIGHT=3;
PROC GPLOT DATA=patient&i GOUT=graphics;
  LABEL Visit="Visit date";
  LABEL BMI="BMI";
  PLOT BMI*Visit/NAME="BMI&i" NOFRAME;
RUN;
PROC GPLOT DATA=patient&i GOUT=graphics;
  LABEL Visit="Visit date";
  LABEL Activity="X Active";
  PLOT Activity*Visit/NAME="Active&i" NOFRAME;
RUN;
PROC GPLOT DATA=patient&i GOUT=graphics;
  LABEL Visit="Visit date";
  LABEL Fruit_Veggies="X Fruit/Veggies";
  PLOT Fruit_veggies*Visit/NAME="Fruit&i" NOFRAME;
RUN;
PROC GPLOT DATA=patient&i GOUT=graphics;
  LABEL Visit="Visit date";
  LABEL SBP="Blood Pressure";
  PLOT SBP*Visit DBP*Visit/NAME="BP&i" OVERLAY NOFRAME;
RUN;

PROC GSLIDE GOUT=graphics NAME="BMItitle&i ";
NOTE HEIGHT=3.5 BOLD JUSTIFY=CENTER "Body Mass Index";
RUN;
PROC GSLIDE GOUT=graphics NAME="Activetitle&i ";
NOTE HEIGHT=3.5 BOLD JUSTIFY=CENTER
"Average Times Physically Active Per Week";
RUN;
PROC GSLIDE GOUT=graphics NAME="Fruittitle&i ";
NOTE HEIGHT=3.5 BOLD JUSTIFY=CENTER
"Average Fruit/Veggie Servings Per Day";
RUN;
PROC GSLIDE GOUT=graphics NAME="Bptitle&i ";
NOTE HEIGHT=3.5 BOLD JUSTIFY=CENTER
"Systolic and Diastolic Blood Pressure";
RUN;

GOPTIONS RESET=ALL DISPLAY HSIZE=9in VSIZE=4.3in;
PROC GREPLAY TC=work.tempcat NOFS;
  TDEF snapshot
  1/LLX=0 LLY=0 ULX=0 ULY=44
    urx=49 ury=44 lrx=49 lry=0
  2/llx=0 lly=0 ulx=0 uly=49
    urx=49 ury=49 lrx=49 lry=0
  3/llx=0 lly=51 ulx=0 uly=95
    urx=49 ury=95 lrx=49 lry=51
  4/llx=0 lly=51 ulx=0 uly=100
    urx=49 ury=100 lrx=49 lry=51
  5/llx=51 lly=51 ulx=51 uly=95
    urx=100 ury=95 lrx=100 lry=51
  6/llx=51 lly=51 ulx=51 uly=100
    urx=100 ury=100 lrx=100 lry=51
  7/llx=51 lly=0 ulx=51 uly=44
    urx=100 ury=44 lrx=100 lry=0
  8/llx=51 lly=0 ulx=51 uly=49
    urx=100 ury=49 lrx=100 lry=0;
  TEMPLATE snapshot;
```

## Looking at the Big Picture – Snapshots of Patient Health in SAS®

```
QUIT;

PROC GREPLAY IGOUT=graphics NOFS TC=work.tempcat TEMPLATE=snapshot;
  TREPLAY 1:"Fruit&i" 2:"Fruitttitle&i " 3:"BMI&i" 4:"BMIttitle&i"
          5:"Active&i" 6:"Activettitle&i" 7:"BP&i" 8:"BPttitle&i";
RUN;
QUIT;

ODS TEXT = "#S={OUTPUTWIDTH=100% JUST=R FONTWEIGHT=BOLD}
Date Modified: #S={}&today ";
ODS &output_type CLOSE;

%END;
%MEND;
%OUTPUT (&maxp);
%MEND;
```

## **ACKNOWLEDGMENTS**

I would like to thank my past supervisor Kathryn Schurr, M.S. for inspiring me to present my snapshot program, and for her continued guidance and support. I would also like to thank Dr. Raymond Neff and Spectrum Health Healthier Communities for giving me the opportunity to further develop my SAS abilities.

## **CONTACT INFORMATION AND BIOGRAPHY**

Your comments and questions are valued and encouraged. Contact the author at:

Author Name: Ruth Kurtycz

Phone: 616-214-9018

Email: kurtyczruth@gmail.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.